

```

imprt = defn model
// Model of Importin-Alpha/Importin-Beta Mediated Nuclear Import
// Authors: Gregory Riddick and Ian G. Macara
// Departments of Biochemistry and Microbiology, University of Virginia

// All concentration values in MicroMolar Units and velocity of a reaction is uM/sec
// Rate constants have been adjusted for micromolar concentration units as follows
// First Order Rate Constants have units of sec(-1) so need no adjustment
// Second Order Rate Constant (Mol(-1)sec(-1)): Divide by 1E6
// Third Order Rate Constant (Mol^2(-1)sec(-1)): Divide by 1E12

vol Cyto, Nuc;

ext GDPy,GDPn,GTPy,GTPn; // GTP/GDP declared as fixed boundary species with unchanging concentrations

// Nuclear species are designated with a lower case "n"
var Rt in Nuc, //RanGTP
    RdNn in Nuc, //RanGTP-NTF2
    Rdn in Nuc, //RanGDP
    Nn in Nuc, //NTF2
    C2n in Nuc, // Endogenous Cargo (ImpB)
    Sn in Nuc, // Cas
    Cn in Nuc, // Endogenous Cargo (ImpA/ImpB)
    An in Nuc, // ImpA
    Bn in Nuc, // ImpB
    ABn in Nuc, // ImpA-ImpB
    ABCn in Nuc, // ImpA-ImpB-Cargo
    RTBn in Nuc, // RanGTP-ImpBeta
    RTsan in Nuc, //RanGTP-Cas-ImpA

    C2Bn in Nuc, //ImpB-Cargo
    ABCFn in Nuc, //ImpA-ImpB-FluorCargo
    ACFn in Nuc, //ImpA-FluorCargo
    CFn in Nuc, //FluorCargo
    C2Fn in Nuc, //FluorCargo
    C2FBn in Nuc, //ImpB-FluorCargo
    RCC1 in Nuc, //RCC1
    RCC1Rd in Nuc, //RCC1-RanGDP
    RCC1R in Nuc, //RCC1-Ran
    RCC1Rt in Nuc, //RCC1-RanGTP
    ABntemp in Nuc, //Intermediate in ImpA/ImpB association
    NRn in Nuc, //Generic Transport Receptor
    RTNRn in Nuc, //RanGTP-Generic Transport Receptor
    ACn in Nuc, //ImpA-Cargo
    RTsACn in Nuc, //RanGTP-Cas-ImpA-Cargo
    RTsn in Nuc, //RanGTP-Cas
    RTsACFn in Nuc; //RanGTP-Cas-ImpA-FluorCargo

//Cytoplasmic species are designated with a lowercase "y"
var Ny in Cyto, //NTF2
    Rd in Cyto, //RanGDP in Cytoplasm
    RdNy in Cyto, //RanGDP-NTF2
    C2y in Cyto, //Endogenous Cargo
    Sy in Cyto, //Cas
    Cy in Cyto, //Endogenous Cargo
    Ay in Cyto, //ImpA
    By in Cyto, //ImpB
    ABY in Cyto, //ImpA-ImpB
    ABCy in Cyto, //ImpA-ImpB-Cargo
    RTBy in Cyto, //RanGTP-ImpB
    RTsAy in Cyto, //RanGTP-Cas-ImpA
    C2By in Cyto, //ImpB-Cargo
    RDpBy in Cyto, //RanGDP-RanBP1-ImpBeta
    RTpBy in Cyto, //RanGTP-RanBP1-ImpBeta
    RTsAPy in Cyto, //RanGTP-Cas-ImpA-RanBP1
    P in Cyto, //RanBP1
    CFy in Cyto, //Fluor Cargo
    ABCFy in Cyto, //ImpA-ImpB-FluorCargo
    C2Fy in Cyto, //Fluor-Cargo
    C2FBy in Cyto, //ImpB-FluorCargo
    RanGap in Cyto, //RanGAP
    ABCYtemp in Cyto, //ImpA-ImpB-Cargo intermediate
    ABCFytemp in Cyto, //ImpA-ImpB-Cargo Fluor intermediate
    C2Bytemp in Cyto, //Cargo Intermediate
    C2FBYtemp in Cyto, //Fluor Cargo Intermediate
    NRy in Cyto, //Generic Receptor
    NRPy in Cyto, //Generic Receptor-RanBP1
    RTNRy in Cyto, //Generic Receptor
    RTsPy in Cyto, //RanGTP-Cas-RanBP1
    RTy in Cyto, //RanGTP in cytoplasm
    RTpY in Cyto, //RanGTP-RanBP1
    RTsY in Cyto; //RanGTP-Cas

// Translocation through the NPC
CFn -> CFy; (0.0005*CFn)-(0.0005*CFy);
C2Fn -> C2Fy; (0.0005*C2Fn)-(0.0005*C2Fy);
C2n -> C2y; (0.001*C2n)-(0.001*C2y);
Cn -> Cy; (0.001*Cn)-(0.001*Cy);
Sn -> Sy; (0.3*Sn)-(0.3*Sy);
RTsan -> RTsAy; (0.2*RTsan)-(0.2*RTsAy) ;
Nn -> Ny; (1*Nn)-(1*Ny);
Rd -> RdNn ; (0.03 * Rd)-(0.03 * Rdn);
Rt -> RTy; (0.03 * Rt)-(0.03 * RTy);
RdNy-> RdNn; (0.5*RdNy)-(0.5*Rdn);
An -> Ay; (0.03*An)-(0.03*Ay);

```

```

Bn -> By; (0.4*Bn)-(0.4*By);
ABn -> ABy; (0.25*ABn)-(0.25*ABy);
C2By-> C2Bn; (0.25*C2By)-(0.25*C2Bn) ;
C2FBy-> C2FBn; (0.25*C2FBy)-(0.25*C2FBn) ;
ABCy -> ABCn; (0.2*ABCy)-(0.2*ABCn);
ABCFy-> ABCFn; (0.2*ABCFy)-(0.2*ABCFn) ;
RtBy -> RtBn; (0.07*RtBy)-(0.07*RtBn) ;
NRy -> NRn; (0.3 * NRy) - (0.3 * NRn);
RtNRn -> RtNRy; (0.07 * RtNRn) - (0.07 * RtNRy);

// Ran Shuttling
Rd + Ny -> RdNy; (10*Rd*Ny)-(1*RdNy);

// RCC1 Mediated GTP Exchange
Rdn + RCC1-> RCC1Rd; (74*Rdn*RCC1)-(55*RCC1Rd);
Rdn + RCC1-> Nn + RCC1Rd; (74*Rdn*Nn*RCC1)-(5.5E-5*Nn*RCC1Rd);
RCC1Rd -> RCC1R + GDPn; (21*RCC1Rd)-(11*RCC1R*GDPn);
RCC1R + GTPn -> RCC1Rt; (0.6*RCC1R*GTPn)-(19*RCC1Rt);
RCC1Rt -> RCC1 + Rt; (55*RCC1Rt)-(100*RCC1*Rt);

// ImpBeta--ImpAlpha--GGNLS complex assembly in Cytoplasm
Ay + By -> ABYtemp ; (0.49*Ay*By)-(0.017*ABYtemp) ;
ABYtemp -> ABy; (7900*ABYtemp)-(0.00025*ABy) ;
ABy + Cy -> ABCYtemp; (0.15*Cy*ABy)-(0.075*ABCYtemp) ;
ABCYtemp -> ABCy; (6400*ABCYtemp)-(0.00044*ABCy) ;

//ImpBeta--ImpAlpha--GGNLS complex disassembly in Nucleus
ACn + Rt -> ACn + RtBn; (0.02*ACn*Rt) - (4.8E-12*ACn*RtBn) ;
Sn + Rt -> RTSn; (0.01*Sn*Rt) - (4.8E-3 * RTSn);
RTSn + ACn -> RTSan + Cn; (0.1*RTSn*ACn) - (1E-10 * RTSan * Cn);
Sn + Rt + ACn -> RTSan + Cn; (1E-7 * Sn * Rt * ACn) - (1E-10 * RTSan * Cn);

// Impalpha--ImpBeta--GGNLS--Fluor Pathway
CFy+ ABY-> ABCFytemp; (0.150000*CFy*ABY)-(0.075*ABCFytemp) ;
ABCFytemp -> ABCFy; (6400*ABCFytemp)-(0.00044*ABCFy) ;
ABCFn + Rt -> ACFn + RTBn; (0.02*ABCFn*Rt) - (4.8E-12*ACFn*RTBn) ;
RTSn + ACFn -> RTSan + CFn; (0.1*RTSn*ACFn) - (1E-11 * RTSan * CFn);
Sn + Rt + ACFn -> RTSan + CFn; (1E-7 * Sn * Rt * ACFn) - (1E-10 * RTSan * CFn);

//Dissassembly of Complexes in the Cytoplasm
RTSay + P -> RTPy + Ay + Sy; (0.30*RTSay*P)-(4E-16*RTPy*Ay*Sy) ;
RTPy -> Rd + P; ((20.1*RanGap*RTPy)/(0.1+RTPy)) ;

//ImpBeta---Cargo Assembly in the Cytoplasm
By+ C2y-> C2Bytemp; (0.490000*By*C2y)-(0.017*C2Bytemp) ;
C2Bytemp -> C2By; (7900* C2Bytemp)-(0.00025*C2By) ;

//ImpBeta---Cargo Dissassembly in the Nucleus
Rt + C2Bn -> C2n + RtBn; (0.02*Rt*C2Bn)-(4.8E-12*C2n*RtBn) ;

//ImpBeta--Cargo--Fluor Pathway
By+ C2Fy-> C2FBytemp; (0.49*By*C2Fy)-(0.017*C2FBytemp) ;
C2FBytemp-> C2FBy; (7900* C2FBytemp)-(0.00025*C2FBy) ;
Rt + C2FBn -> C2Fn + RtBn; (0.02*Rt*C2FBn)-(4.8E-12*C2Fn*RtBn) ;

//Carrier Cycling
An + Bn -> ABntemp ; (0.490000*An*Bn)-(0.017*ABntemp) ;
ABntemp -> ABn; (7900*ABntemp)-(0.00025*ABn) ;
Rt + Br -> RTBn ; (0.096*RTBn)-(4.8E-6*RTBn);
Rt + ABn -> An + RTBn; (0.02 *Rt*ABn)-(4.8E-12*An*RTBn);

//Cycling of Other Transport Receptors
NRn + Rt -> RTNRn; (0.096*Rt*NRn)-(4.8E-6*RTNRn);
RTNRy + P -> RTNRPy; (0.300000*RTNRy*P)-(0.0004*RTNRPy);
RTNRPy -> Rd + P + NRy; ((21.2 * RanGap * RTNRPy)/0.1 + RTNRPy);

//Reactions Due to RanGTP Leakage into the cytoplasm
RTy + By -> RTBy; (0.096*RTy*By)-(4.8E-6*RTBy);
Sy + RTy + Ay -> RTSay; (1E-7 * Sy * RTy * Ay) - (1E-10 * RTSan);
RTy -> Rd ; ((20.1*RanGap*RTy)/(0.7+RTy));
RTy + P -> RTPy; (0.3 * RTy * P) - (4E-4 * RTPy);

//Total Cargo Concentration in Nucleus
ACT = (CFn + ABCFn + ACFn);

end;

```

```

imprt.Cyto = 1.55; // Cytoplasmic Compartment volume
imprt.Nuc = 0.45; // Nuclear Compartment volume

```

```

cf = (imprt.Nuc + imprt.Cyto)/imprt.Cyto; // Used for calculating starting cytoplasmic concentration from whole cell concentration
cn = (imprt.Nuc + imprt.Cyto)/imprt.Nuc; // Used for calculating starting nuclear concentration from whole cell concentration

```

```

imprt.GTPn = 470;
imprt.GDPn = 1.6;
imprt.GTPY = 470;
imprt.GDPY = 1.6;
imprt.Ny = 0.6 * cf;
imprt.Rd = 5 * cf ;
imprt.RanGap = 0.5 * cn;
imprt.RCC1 = 0.25 * cn;
imprt.Sy = 3 * cf ;
imprt.Ay = 1 * cf ;
imprt.By = 3 * cf ;
imprt.P = 2 * cf ;
imprt.C2y = 1;
imprt.Cy =10;
imprt.NRy = 3.6;

//Set everything else to ~0
imprt.RdnN =1E-6;
imprt.Rt =1E-6;
imprt.Nn = 1E-6;
imprt.Rdn = 1E-6;
imprt.RdnNy =1E-6;
imprt.RCC1Rd = 1E-6;
imprt.RCC1R = 1E-6;
imprt.RCC1Rt = 1E-6;
imprt.C2n = 1E-6;
imprt.Sn = 1E-6;
imprt.Cn = 1E-6;
imprt.Nn = 1E-6;
imprt.An = 1E-6;
imprt.Bn = 1E-6;
imprt.ABn = 1E-6;
imprt.ABCn = 1E-6;
imprt.RtBn = 1E-6;
imprt.RtSAn = 1E-6;

imprt.An =1E-6;
imprt.Bn =1E-6;
imprt.ABy = 1E-6;
imprt.ABytemp = 1E-6;
imprt.ABn =1E-6;
imprt.ABCy = 1E-6;
imprt.ABCytemp = 1E-6;
imprt.ABCn = 1E-6;
imprt.RtBy = 1E-6;
imprt.RtBy =1E-6;
imprt.RtSay = 1E-6;
imprt.RtSay =1E-6;
imprt.ACn = 1E-6;

imprt.C2By = 1E-6;
imprt.C2Bn = 1E-6;
imprt.C2n = 1E-6;
imprt.RtPBy = 1E-6;

imprt.CFy = 1E-6;
imprt.CFn = 1E-6;
imprt.ACFn = 1E-6;
imprt.ABCFy = 1E-6;
imprt.ABCFn = 1E-6;

imprt.C2Fn = 1E-6;
imprt.C2FBn = 1E-6;
imprt.C2Fy = 1E-6;
imprt.C2FBy = 1E-6;
imprt.C2FBtemp = 1E-6;
imprt.C2Bytemp = 1E-6;
imprt.ABntemp = 1E-6;

imprt.NRn = 1E-6 ;
imprt.RtNRn = 1E-6;
imprt.NRPY = 1E-6;
imprt.RtNRy = 1E-6;
imprt.Rty = 1E-6;
imprt.RtSm = 1E-6;
imprt.RtSACn = 1E-6;
imprt.RtSACFn = 1E-6;
imprt.RtSPy = 1E-6;
imprt.RtSy = 1E-6;

imprt.sim.atol = 1E-06;
imprt.sim.rtol = 1E-06;

// Take system to Steady-State
m = imprt.sim.eval(0, 1000, 50, [<imprt.Time>,<imprt.Cn>]);

// Add GGNLS Cargo
imprt.CFy = 3;
imprt.RCC1 = imprt.RCC1 + 20;

// Simulate GGNLS Cargo Import
p = imprt.sim.eval(0, 1800, 91, [<imprt.Time>,<imprt.ACT>]);
graph(p);

```